This worksheet demonstrates how Maple can be used to find regular expansions for 2-point boundary value problems with Dirichlet boundary conditions. The problem must have the form

$$\text{ODE}=f\left(\frac{d^2}{dx^2}\,y(x),\ \frac{d}{dx}\,y(x),\ y(x),\ x,\ \varepsilon\right)=0$$

where $y(0)=A(\varepsilon)$ and $y(1)=B(\varepsilon)$ for x in (0,1). As the worksheet is written, the first two terms in the expansion

$$y(x)=y_0(x)+\varepsilon\,y_1(x)+O(\varepsilon^2)$$

are found. At the end, if the (exact) problem can be solved by Maple, graphs of the true solution and asymptotic approximations are compared.

To use the worksheet, define ODE, A and B below and then just keep hitting return.

```
>
> ODE:=diff(y(x),x$2)+epsilon*diff(y(x),x)-y(x)+epsilon*y(x)+epsilon
  *diff(y(x),x$2);
```

$$ODE := \left(\frac{d^2}{dx^2}\,y(x)\right)+\varepsilon\left(\frac{d}{dx}\,y(x)\right)-y(x)+\varepsilon\,y(x)+\varepsilon\left(\frac{d^2}{dx^2}\,y(x)\right)$$

```
> A:=1-epsilon;B:=0;
```

$$A := 1-\varepsilon$$

$$B := 0$$

```
>
```

The following procedures are used to make expansions of y(x) and expand the differential equation into its respective orders.

```
> N:=2:
> makefnexpansion:=proc(y,x,epsilon,N)
>           local i,expansion;
>           expansion:=0;
>           for i from 0  by 1 to N do
  expansion:=expansion+epsilon^i*y[i](x)  od;
> end:
> makefnexpansion(y,x,epsilon,N);
```

$$y_0(x)+\varepsilon\,y_1(x)+\varepsilon^2\,y_2(x)$$

```
> EXPANDFN:=proc (expression,y,x,epsilon,N)
>            local z0,z1,z2,z3;
>            z0:=makefnexpansion(y,x,epsilon,N);
>            z1:=subs(y(x)=z0,expression);
>            z2:=series(z1,epsilon=0,N+1);
>            z3:=collect(z2,epsilon);
```

```
> end:
>
>
```

First, we determine the $O(\varepsilon^i)$ differential equation for $y_i(x)$:

```
> for i from 0 by 1 to N do
  DEqn[i]:=coeff(EXPANDFN(ODE,y,x,epsilon,N),epsilon,i)=0 od;
```

$$DEqn_0 := (D^{(2)})(y_0)(x) - y_0(x) = 0$$

$$DEqn_1 := (D^{(2)})(y_1)(x) + y_0(x) + D(y_0)(x) + (D^{(2)})(y_0)(x) - y_1(x) = 0$$

$$DEqn_2 := (D^{(2)})(y_2)(x) + y_1(x) + D(y_1)(x) + (D^{(2)})(y_1)(x) - y_2(x) = 0$$

Next, we determine the $O(\varepsilon^i)$ boundary conditions for $y_i(x)$. The left boundary conditions are BC[i,0]:

```
> for i from 0 by 1 to N do
  BC[i,0]:=y[i](0)=coeff(series(A,epsilon,N),epsilon,i) od;
```

$$BC_{0,\,0} := y_0(0) = 1$$

$$BC_{1,\,0} := y_1(0) = -1$$

$$BC_{2,\,0} := y_2(0) = 0$$

```
> for i from 0 by 1 to N do
  BC[i,1]:=y[i](1)=coeff(series(B,epsilon,N),epsilon,i) od:
```

Now we find the solution Y[0] of the leading problem for $y_0(x)$:

```
> Y[0]:=rhs(simplify(dsolve({DEqn[0],BC[0,0],BC[0,1]},y[0](x))));
```

$$Y_0 := -\frac{\mathbf{e}^x - \mathbf{e}^{(2-x)}}{\mathbf{e}^2 - 1}$$

This must be substituted into the $O(\varepsilon)$ problem for $y_1(x)$ before Maple can solve it:

```
> NDEqn[1]:=subs((D@@2)(y[0])(x)=diff(Y[0],x$2),DEqn[1]):
> NDEqn[1]:=subs(D(y[0])(x)=diff(Y[0],x),NDEqn[1]):
> NDEqn[1]:=subs(y[0](x)=Y[0],NDEqn[1]);
```

$$NDEqn_1 := (D^{(2)})(y_1)(x) - \frac{2\,(\mathbf{e}^x - \mathbf{e}^{(2-x)})}{\mathbf{e}^2 - 1} - \frac{\mathbf{e}^x + \mathbf{e}^{(2-x)}}{\mathbf{e}^2 - 1} - y_1(x) = 0$$

Now we solve the problem for y[1](x): (Beware, solutions can be messy and long)

```
> Y[1]:=rhs(simplify(dsolve({NDEqn[1],BC[1,0],BC[1,1]},y[1](x))));
```

$$Y_1 := -\frac{1}{2}\,\frac{-\mathbf{e}^{(-x+4)}\,x+2\,\mathbf{e}^{(-x+4)}+2\,\mathbf{e}^{(x+2)}-3\,\mathbf{e}^{(x+2)}\,x-6\,\mathbf{e}^{(2-x)}+\mathbf{e}^{(2-x)}\,x+3\,\mathbf{e}^{x}\,x+2\,\mathbf{e}^{x}}{(\mathbf{e}^{2}-1)^{2}}$$

Providing the original problem you input can be solved by Maple, the next procedures create plots comparing the exact solution to the $y_0(x)$ and the higher order correction $y_0(x)+\varepsilon\, y_1(x)$

The value "eps" below is some numerical value of epsilon for your comparison.

```
> Compare0:=proc(ODE,A,B,eps,Y0)
> local ODEX,BCL,BCR,EXACT;
> ODEX:=subs(epsilon=eps,ODE);
> BCL:=y(0)=subs(epsilon=eps,A);
> BCR:=y(1)=subs(epsilon=eps,B);
> EXACT:=rhs(dsolve({ODEX,BCL,BCR},y(x)));
> plot({EXACT,Y0},x=0..1);
> end:
>
```

First: the leading approximation of the solution

```
> Compare0(ODE,A,B,1/10,Y[0],Y[1]);
```




```
> Compare1:=proc(ODE,A,B,eps,Y0,Y1)
> local ODEX,BCL,BCR,EXACT;
> ODEX:=subs(epsilon=eps,ODE);
> BCL:=y(0)=subs(epsilon=eps,A);
> BCR:=y(1)=subs(epsilon=eps,B);
> EXACT:=rhs(dsolve({ODEX,BCL,BCR},y(x)));
> plot({EXACT,Y0+eps*Y1},x=0..1);
> end:
```

Then, the higher order correction. If you don’t see two curves its probably because the higher order correction is so accurate.

```
> Compare1(ODE,A,B,1/2,Y[0],Y[1]);
```


0.5
0.4
0.3
0.2
0.1
0
0.2
0.4
0.6
0.8
1
x

>

>